# Maintainer: Balló György <ballogyor+arch at gmail dot com>

pkgname=gnome-multi-writer
pkgver=3.35.90
pkgrel=1
pkgdesc='Write an ISO file to multiple USB devices at once'
arch=('loong64' 'x86_64')
url='https://wiki.gnome.org/Apps/MultiWriter'
license=('GPL')
groups=('gnome-extra')
depends=('dconf' 'gcc-libs' 'glib2' 'glibc' 'gtk3' 'hicolor-icon-theme' 'libcanberra' 'libgudev'
         'libgusb' 'polkit' 'udisks2')
makedepends=('appstream-glib' 'docbook-sgml' 'docbook-utils'  'meson' 'perl-sgmls')
source=("https://download.gnome.org/sources/$pkgname/${pkgver%.*}/$pkgname-$pkgver.tar.xz")
sha256sums=('b7fe3667e23402ee1d5b01c5d271dd6475a79e4a70843f0e3c97f5dca4fc6f1f')

build() {
  arch-meson $pkgname-$pkgver build
  meson compile -C build
}

package() {
  meson install -C build --destdir "$pkgdir"
}
